

Chapitre 2 Sas Windows : quelques repères techniques cachés

L'utilisation de Sas, sous Windows, active en fait des opérations d'arrière-plan, qui ont ou auraient pu être menées à bien : il s'agit maintenant de les signaler et les expliciter.

Elles concernent

- la création et l'utilisation de tables Sas sur le disque du micro ou du serveur accessible en réseau
- l'exécution automatique d'un fichier de configuration du logiciel (SASV9.CFG) selon votre appel
- l'exécution automatique d'un second fichier de démarrage (AUTOEXEC.SAS)
- la sauvegarde de vos habitudes et usages de travail en Sas, propre à votre poste de travail ou à vous-même (les catalogues PROFILE)
- l'utilisation d'un espace temporaire de travail (la WORK)

2.1 Etablir la liaison entre des fichiers et Sas Windows (objets internes et fichiers externes)

2.1.1 Le libname : accès aux objets Sas

Un LIBNAME établit un lien entre un *libref* (une étiquette logique choisie par vous) et un répertoire (ou sous-répertoire) **physiquement préexistant** sur disque. La table Sas permanente ira donc s'écrire dans le répertoire référencé par le *libref*. Le suffixe lié au du nom physique d'une table Sas 8 ou 9 sera **.SAS7BDAT**.

Pour décochez le masquage des extensions de votre Explorateur aller à *Outils /Options des dossiers onglet affichage*

Masquer les extensions des fichiers dont le type est connu en W7 cherchez « masquer extensions »

Ainsi donc : LIBNAME ADMIN "D:\INED\ANDRE2\DON\" ; DATA ADMIN.TABLE1; ...

fera référence au contenu du futur fichier de nom : d:\ined\andre2\don\table1.sas7bdat et si ce fichier existe déjà, il sera écrasé et remplacé sauf si l'option replace= a été utilisée..

Par défaut, une librairie/bibliothèque est pourvue d'un moteur ou engine : En version 9, ce sera par défaut :

LIBNAME ADMIN V9 "D:\USER\ANDRE2\DON\" ;

ATTENTION : Il existe aussi des librairies à engine mixte sans mention explicite d'engine, où nous rencontreront des tables 6.12, 8 et 9. Une librairie mixte sans engine deviendra v6, s'il y a une vieille table v6 comme **unique** objet sas dans ce répertoire! Convertissez-là vite en v9. !!

En Sas Windows, l'allocation de l'espace des fichiers (tables, catalogues,...) est automatiquement gérée par le système d'exploitation. On parlera d'allocation dynamique. Que ceci ne vous dispense pas, en cas de grosse manipulation ou en cas de partitionnement de disque, de calculer l'espace des grosses tables à créer, de faire le ménage préalable et d'effectuer la compression *interne* à SAS des tables. Zipper rend les tables Sas inaccessibles. Par contre, vous pouvez vous frotter prudemment aux compressions système.

Dans une librairie Sas, on peut trouver d'autres composantes (moins utilisées) que des tables Sas :

Fichiers en version 8 et 9 de SAS	Accès Direct Windows/Linux	Séquentiel Linux/unix	Éléments du Système SAS**
Programme SAS	.sas	.sas	-
Setinit (licence)	.sss/.sas*	.sss/.sas*	-
Log SAS	.log	.log	-
Fichier output	.lst	.lst	-
Table SAS	.sas7bdat	.sas7sdat	DATA
Catalogue SAS	.sas7bcat	.sas7scat	CATALOG
Index SAS	.sas7bndx	.sas7sndx	INDEX
Programme stocké avec une étape data	.sas7bpgm	.sas7spgm	PROGRAM
Vue SAS	.sas7bview	.sas7svew	VIEW
Fichier de description d'accès	.sas7bacs	.sas7sacs	ACCESS
Fichier d'audit	.sas7baud	.sas7saud	AUDIT
Base de données multi-dimensionnelle	.sas7bmdb	.sas7smdb	MDDB
Fichier system ODS	.sas7bods	.sas7sods	SASODS
Fichier de données Data Mining	.sas7bdmd	.sas7sdmd	DMDB
Fichier d'éléments stockés	.sas7bitm	.sas7ssitm	ITEMSTOR

Fichier utilitaire	.sas7but1	.sas7sut1	UTILITY
Fichier utilitaire permanent	.sas7bputy	.sas7sputy	PUTILITY
Fichier de sauvegarde	.sas7bbak	.sas7sbak	BACKUP

VARIATIONS SUR LES LIBNAMES

Un LIBNAME ouvrant sur plusieurs répertoires est autorisé; le premier reste prioritaire à l'écriture.

```
LIBNAME ADMIN V9 ("D:\USERS\ANDRE\","G:\DONNEES\");
```

```
LIBNAME ENSEMB (LREF1 LREF2) ; *deux librefs déjà existants ;
```

L'option ACCESS= dans l'instruction LIBNAME. peut prendre la valeur READONLY pour ne fonctionner qu'en mode lecture avec interdiction d'écriture et donc respect des données d'autrui.

```
LIBNAME INDPRIX "F:\INED\" ACCESS=READONLY ;
```

Pour les besoins d'une grosse application, il est possible d'utiliser une librairie au contenu temporaire:

```
LIBNAME TRANSIT "D:\USERS\TRAV\" ACCESS=TEMP ;
```

Si vous avez tendance à oublier les ; finaux des instructions Sas, l'option `reempty=no` évitera de remplacer une table existante par une table de même nom vide. L'option `compress=yes|binary` assurera une compression de toutes les tables. *Yes* en cas de majorité de variables caractères, mais *Binary* en cas de très grosse majorité de variables numériques (*analysez votre log svp !*)

```
LIBNAME PROTECT "D:\TRAVAIL\" REEMPTY=NO COMPRESS=YES;
DATA PROTECT.A SET PROTECT.B;RUN;
```

On peut aussi souhaiter utiliser un nom de table simple (sans libref) et néanmoins conserver celle-ci comme permanente grâce à l'instruction suivante :


```
LIBNAME USER "D:\USERS\PERM\" ; DATA A;... ; Run;
```

L'utilisation du USER fait que la table A ira s'écrire dans le répertoire indiqué de façon permanente et seule, une référence explicite à la work (`data work.temp;`) permettra d'écrire une table réellement temporaire dans la work et qui sera effacée à la fermeture de Sas!

Créer un nouveau libname peut aussi se faire de façon interactive par une nouvelle fenêtre appelée par le bouton



. On peut aussi l'appeler par la rubrique *NewLibrary/Nouvelle Bibliothèque* du menu déroulant Tools/Outils, ou dans fenêtre Explorer/Libraries objet actif avec un clic droit et enfin par une commande DMLIBASSIGN



```
ACCESS      SASECRSP
BLOOMBRG    SASEFAME
BMDP        SASEHAVR
CVP         SASTS
DB2         SPDE
EXCEL       SPSS
INFOMAPS    SYBASE
META        TERADATA
MYSQL       V6
NEOVIEW     V604
NETEZZA     V9
ODBC        XML
OLEDB       XML92
ORACLE      XPORT
OSIRIS
PCFILES
REMOTE
REMOTE8
REUTERS
```

avec ces engine

Cocher « Enable at startup », créera la librairie et son libref/étiquette à chaque démarrage de Sas, car ce *libref* sera mémorisé dans la clé du registre SAS [CORE\OPTIONS\LIBNAMES].

Pour se libérer de ce processus associatif au démarrage et nettoyer cette clé du registre, il suffit de supprimer le *libref* par un `libname libref clear ;` instruction généralement utile pour se défaire d'un libname.

Prenez cependant l'habitude de mettre les libname et filename en tête de programme ! car dans 5 ans, vous ne vous souviendrez d'aucune de vos associations automatiques implicites.

Si vraiment vous y êtes opposé, car vous voulez en fait changer le lieu par défaut de Sas pour ouvrir vos programmes, essayer la commande DLGCDIR.

Je serais bien plus tenté d'utiliser cette fenêtre pour créer un libname spécial du genre Excel ou Xml ou Spss, car les informations sollicitées en supplément du path/chemin sont plus détaillées, mais sans jamais cocher le enable at startup.

De façon temporaire, vous pouvez faire usage de cette forme de libname

```
LIBNAME POURQUOIPAS ".";
```

Vous aurez fixé comme lieu pour cette bibliothèque, le répertoire indiqué dans la Status Bar en bas de l'espace visuel Sas.

Existe aussi la possibilité de travailler avec une librairie en mémoire Libname a "___" memlib ; (bon en 64bit car mémoire extensible)

Enfin, il existe, (pour les utilisateurs avertis), une dernière façon de fixer un répertoire permanent pour son travail Sas, en s'appuyant sur une variable d'environnement Sas, créée dans le fichier config Sasv9.cfg (voir en 2.2)


2.1.2 Le filename : un des accès aux fichiers externes

Une instruction FILENAME associe un *fileref* (une étiquette logique) avec un fichier externe (généralement de format texte ou propre à un autre logiciel) entièrement nommé et utilisé en lecture ou en écriture.

```
FILENAME TADONNE "D:\USERS\ANDRE2\DON.DAT" ;
DATA A; INFILE TADONNE . . . ; . . .
```

Il est possible d'appeler une fenêtre permettant l'établissement du filename par la commande DMFILEASSIGN ou par l'item *NewFileShortcut /Nouveau Raccourci* du Menu Tools/Outils, ou enfin par un clic droit sur l'icône File Shortcut de l'Explorateur Sas

		<p>Signification des filenames icônifiés</p> <table> <tr> <td>Rlink</td> <td>icône du sas connect</td> </tr> <tr> <td>H</td> <td>icône d'un accès DISK</td> </tr> <tr> <td>Peche</td> <td>icône d'un accès URL</td> </tr> <tr> <td>Allfil</td> <td>icône d'un accès PIPE</td> </tr> <tr> <td>Ftpin</td> <td>icône d'un accès FTP</td> </tr> </table>	Rlink	icône du sas connect	H	icône d'un accès DISK	Peche	icône d'un accès URL	Allfil	icône d'un accès PIPE	Ftpin	icône d'un accès FTP
Rlink	icône du sas connect											
H	icône d'un accès DISK											
Peche	icône d'un accès URL											
Allfil	icône d'un accès PIPE											
Ftpin	icône d'un accès FTP											



Le *enable at startup* existe ici aussi et évoque les mêmes réserves de ma part. Absence du LRECL !

Il faut bien comprendre qu'un filename donne accès à du contenu généralement texte (TXT) qu'il soit du type données (.dat ou .cport), programme (.sas), résultat (.lst) ou journal d'exécution (.log).

Les méthodes d'accès spécialisé peuvent s'avérer intéressantes.
Méthode *email* avec fileref/étiquette *new*

Permet la soumission d'un programme « avertisseur » en cas d'_error_ dans Sas

<pre> data aa;set xx; *test car xx inexistant; run; %macro sendmail; %if &syserr ne 0 %then %do; data _null_; FILE new; PUT 'à retravailler '; PUT "Non résolu &sysdate &systime"; run; %end; %mend sendmail; %sendmail; </pre>	<p>Equivalent à cette instruction</p> <pre> *filename new email to="qwo-62fa@myamail.com" su="à nouveau une erreur"; </pre> <p>mais sous réserve que, dans le config sasv9.cfg sollicité, vous ayez bien ce genre de renseignements :</p> <ul style="list-style-type: none"> -EMAILSYS SMTP -EMAILHOST mailserv.ined.fr -EMAILPORT 25 -EMAILAUTHPROTOCOL=none <p>(-emailid adresse semble ne plus être nécessaire)</p>
---	---

Et la méthode *url* (voir un peu plus loin)

Par contre, dans un programme, il vous faudra écrire du code du genre suivant pour envoyer un résultat :

```

FILENAME MAIL EMAIL
SUBJECT = "Participants au Congrès"
FROM = " qwo@myail.com" TO="wielki@ined.fr"
REPLYTO="org@meeting.fr"
TYPE = "text/html"
ATTACH=("d:\test.xls" content-type="text/plain");
ods listing close;
ODS HTML BODY=MAIL;
Title ;
PROC SQL;SELECT name FROM sashelp.class; QUIT;
ods html close;
ods listing;

```

La possibilité d'un filename ne mentionnant qu'un répertoire (ou sous-répertoire) est autorisée, mais fonctionne de la manière suivante quant à l'utilisation du fileref/étiquette:


- dans une instruction INFILE ou FILE en supposant un suffixe **.DAT** pour les fichiers lus ou créés

```
FILENAME MONPROG "D:\USERS\ANDRE\ETUD\" ;
DATA PERSINFO ; INFILE MONPROG(INFO) ;
INPUT NOM $15. PRENOM $10. DATENT DDMMYY6. GRADE $5.; RUN;

DATA _NULL_ ; SET PERSINED; FILE MONPROG(INED) NOPRINT;
PUT NOM $15. PRENOM $10. DATENT DDMMYY6. GRADE $5.;
RUN;
* LES NOMS DE FICHIERS SONT INFO.DAT ET INED.DAT ;
```

- dans une commande include ou file utilisée sur la fenêtre Pgm en supposant un suffixe **.SAS**

```
FILENAME MONPROG "D:\ASTUCES\" ;
```



permet de lancer un **FILE** MONPROG (CAPS) en commande de la fenêtre pour sauvegarder le programme courant dans le fichier CAPS.SAS du répertoire référencé dans le filename ou un **INCLUDE** MONPROG (DISJONC) pour appeler dans la fenêtre, le programme Sas stocké dans DISJONC.SAS du même répertoire.

- dans une instruction %INCLUDE en supposant un suffixe **.SAS**

```
FILENAME TEST "D:\ASTUCES\" ;
%INCLUDE TEST(DISJONC) ;
```

Cet appel à l'exécution des instructions Sas stockées dans DISJONC.SAS du répertoire mentionné est en général méconnu.

```
%INCLUDE monprog(capslow)/source2;
```

Cet appel à l'exécution s'opère avec l'insertion visuelle dans la log du code Sas

La concaténation de fichiers est autorisée et peut s'écrire par exemple comme suit :

```
FILENAME DONTOU ("PARIS.DAT" "BANLIEUE.DAT") ;
OU FILENAME TOUARR "ARR*.DAT" ;
* POUR ARR01.DAT ARR02.DAT ... ARR20.DAT DE PARIS ;
```

Pour ceux qui l'ignorent, il existe dans Sas, la possibilité (peu usitée il est vrai) de stocker des programmes dans un catalogue Sas sous la forme d'objets Source. Il sera ensuite également autorisé d'utiliser l'instruction Filename sur un catalogue sous réserve de préciser l'option *catalog* :

```
FILENAME MONCAT CATALOG "SASUSER.MONCAT.MONPGM.SOURCE" ;
%INCLUDE MONCAT ;

OU
FILENAME MONCAT CATALOG "SASUSER.MONCAT" ;
%INCLUDE MONCAT (MONPGM) ;
```

Ceci permet en quelque sorte l'accès direct aux catalogues sas de programmes sources.

En voici un autre exemple à propos de macros stockées sous la forme .SOURCE dans un catalogue (ici la macro %special stockée dans le catalogue personnel MACRPERS situé dans la SASUSER).

```
FILENAME MM CATALOG "SASUSER.MACRPERS" ;
OPTIONS SASAUTOS=MM MAUTOSOURCE ;
%special;
```

Mentionnons ici aussi l'existence d'une option du filename **ftp** permettant d'atteindre directement un fichier texte situé ailleurs sur un serveur acceptant votre sollicitation

Exemple 1 : lecture directe d'un fichier texte situé sur Linux dans un répertoire qui n'est pas de mon espace réservé, mais aux droits de lecture linux non restreints (764 par exemple rw- rw- r--)

```
filename ftpin ftp '/home/users/lad_m/fond/s1986.txt' recfm=f
      host="margaux.recherche.ined.fr" user="wielki"
      pass="xxxxxxxx" ;
data test;
infile ftpin dlm='09'x dsd lrecl=200 pad termstr=nl;
input num id dep segment x y;
run;
```

Il n'y a plus à partir de la version 9.2 d'option OEM possible pour le filename, les choses devant se régler par l'environnement NLS de Sas. Par contre, Sas vous autorise à utiliser l'option encoding= "utf-8" ou "utf161" dans le Filename, pour affronter des textes formatés selon un encodage international.

Exemple 2 : en Windows, lecture directe d'un programme concocté par Sas Institute et dépôt/écriture correct de celui-ci dans un sous répertoire de mon espace réservé sur serveur Margaux sous linux

```
* essai      fichier TEXTE  pgm  sas extérieur vers linux ;
filename ftpsas ftp 'SASInstallReporter2.sas' recfm=f
      host="ftp.sas.com" user="anonymous" pass="wielki@ined.fr"
      cd="techsup/download/base/" rcmd="type a"
      debug;
filename ftplin ftp '/home/users/wielki/temp/reportermodule.sas' recfm=f
      host="margaux.recherche.ined.fr" user="wielki"
      pass="xxxxxxxx" rcmd='type a' debug;
      * type a spécifie bien des 2 côtés que l'on a affaire avec du texte ascii et non du code binary ;
data _null_;
  infile ftpsas;
  file ftplin;
  input;
  put _infile_;
run;
```

Quelques autres usages du filename sont possibles grâce à l'option url.

Ex 1: lecture de données médicales à partir d'un site de statistiques en ligne à une adresse url grâce à une étape data -ici élémentaire- !

```
filename cigar url "http://lib.stat.cmu.edu/DASL/Datafiles/cigcancerdat.html";
data cc;
infile cigar dlm="09"x firstobs=38 trunccover;
length state $2;
input STATE $ CIG BLAD LUNG KID LEUK ;
if state ne "</" then output;
run;
```

Ex 2 : Récupération d'une macro sur un site distant (Michael Friendly et ses macros statistiques au Canada voir page 114) pour exécution avec visualisation du code

```
filename fin URL 'http://euclid.psych.yorku.ca/ftp/sas/vcd/macros/label.sas';
%include fin/source2;
filename fin clear;
```

Ex 3 : Récupération pour utilisation immédiate de la dernière version en ligne du tagset Excelxp chez SAS dans ma librairie de templates/modèles personnelle *sans* son très long listage !

```
libname my "D:\My SAS Files\9.2\SG";
ods path my.templ(update) sashelp.tmplmst(read);
filename tagset url 'http://support.sas.com/rnd/base/ods/odsmarkup/excltags.tpl';
%include tagset;
quit;
ods listing close;
ods tagsets.excelxp ... ; voir ce sujet au chapitre 9.8
```

Ex 4 : Lecture d'un fichier texte ascii sur le serveur d'images copie de cdrom/dvd Pecharmant

```
options errors=1;*vu ce fichier ancien je teste !;
filename in url 'http://pecharmant.ined.fr:80/Fig_Log/Logement84/Ascii/pror84.txt'
user='INED/wielki' pass="xxxxxx" debug;
data propriete;
infile in firstobs=2 lrecl=250 pad ;
input QUEST SPP SPL SPI SPH SAA SAA1 FBP FEP1 FEP2 FCC FEP FSF FNP FMP FCM FCCM
FECM FTCM;
run;
```

Mentionnons la lecture expérimentale directe d'un fichier de données externes (pas une table sas !) contenu dans un fichier .zip. à l'aide de l'engine saszipam

(Utile pour l'enseignant ou un manuel de cours : il ne faudra pas dézipper le fichier)

```
filename x saszipam 'd:\everittbook\everittbook.zip';
data knots;
infile x(knots.dat) ;
input a $ b c d ;
run;
```

Une autre mine d'or est le recours à l'option pipe

Sous Windows (mais aussi sous Linux), il est possible de lire le résultat d'une commande passée au système d'exploitation ligne à ligne et d'en picorer les informations, grâce à un 'pipe' (tunnel de communication entre deux processus). Pas besoin de changer les options xwait etc.

Pour approfondir servez-vous de <http://www2.sas.com/proceedings/forum2008/092-2008.pdf>

“Check out These Pipes: Using Microsoft Windows Commands from SAS®” de Brian Varney

Exemple 1 : cherche à repérer sur le disque d:\ tous les fichiers sas .sd2 pour ENFIN les convertir en les mettant à niveau pour la dernière version au suffixe .sas7bdat.

```
filename out1 "d:\sasanc\liste_sd2.txt";
filename in pipe "dir d:\*.sd2 /s ";
data in (keep=nomrep) ;
length nomrep $50;
infile in length=ll;
file out noprint;
input @2 test $2. @ ;lrr=ll-15;
if test="R" then do;
input @16 nomrep $varying. lrr;
output;
put nomrep $50.;
end;
else input;
run;
```

Exemple 2 : ayant désigné un répertoire, je souhaite en connaître les noms de fichiers

(début de solution car il reste le pb des caractères accentués si vous en avez mis, mais j'ai une solution + loin)

```
%let repertoire= %str(d:\public\avous);
/* %let fref=%unquote(%str('%')dir /Q %str('%')&repertoire.\*. * %str('%')%str('%'));
filename allfil pipe &fref.; ancienne manière */
```

```

%let fref=%sysfunc(quote('dir /Q "&repertoire.\*.*'));
filename allfil pipe %sysfunc(dequote(&fref));
data nomspublic(label="&repertoire" drop=bin);
  length date heure bin $ 15 prop $25 nom $50;
  infile allfil truncover end=eof ;
  input date :$10. heure :$10. bin prop nom &:$50.;
  if _n_ >6;
  if bin ne "<REP>";
run;

```

Exemple 3 : lister les fichiers y compris ceux des sous-répertoires sur la log

```

%let currrdir = d:\public\avous;
data dir;
  length command $256;
  command = catx(' ', 'dir /Q/S', quote(symget('currrdir')));
  infile dummy pipe filevar=command end=eof
    length=length truncover;
  do while(not eof);
    input info :$256.;    put _infile_;
  end;
  stop;
run;

```

Exemple 4 : Récupérer toutes les tables de la work en les copiant vers un répertoire où elles resteront à la fermeture de Sas

```

%let work = %sysfunc(pathname(work));
filename copy pipe "copy ""&work\*.sas7bdat"" ""d:\temp""";
data _null_ ;
  infile copy;
  input;    put _infile_;
run;

```

Exemple 5 : Quelles sont les ressources déjà montées sur le PC c'est à dire les lettres c : d : etc

```

filename montage pipe drivemap;
data map;
  infile montage; input ressource $ ;    put ressource;
run;

```

Exemple 6 : capture du contenu des variables d'environnement Sas (merci Richard de Venezia)

```

data envvars;
  length name $32 value $1024; command = 'SET';
  infile dummy pipe filevar=command dlm='=' truncover end=eof lrecl=2048;
  do while(not eof);
    input name value $char1024. ;
    output;
  end;
  stop;
  keep name value;
run;

```

Exemple 7 : Retour sur le problème de Dir qui retourne du texte avec des accents visuellement transformé car usant d'un encodage spécial.

Comme une spécification d'encodage dans les instructions filename ftp ou pipe ne sont pas acceptées, voici une solution générale de transformation d'un fichier texte « Dos » en fichier txt courant. La création du fichier s'est faite par une commande Dos utilisant la commande Dir et ensuite Sas a pris le relais.

```

options noxsync noxwait;
x 'for /d %d in (d:\) do dir /? > d:\Temp\syntaxeDir.txt';
filename file "d:\temp\syntaxeDir.txt" encoding=pcoem858;
filename ok "d:\temp\syntaxeDirOK.txt";
data _null_ ;
  infile file ;
  file ok ;
  input ;    put _infile_;
run;

```


Cependant pour la liste des noms de fichiers d'un répertoire, une autre solution existe

```
%let rep=d:\public\avous\eva\;
data files;
  length file $100.;
  rc = filename("d", "&rep");
  fh = dopen("d");
  do i=1 to dnum(fh);
    file=dread(fh,i);
    output;
  end;
  rc=dclose(fh);
run;
```

Rappel de la commande Dir (help dir)

Attention en W7, le pipe ne marche que si vous avez ouvert Sas à partir du déroulement de « Tous les programmes3 et non pas à partir d'une icône du Bureau !!!

Affiche une liste de fichiers et de sous-répertoires dans un répertoire.

Ajout W7

```
DIR [lecteur:][chemin][nom_de_fichier] [/A[:]attributs]] [/B] [/C] [/D] [/L]
[/N] [/O[:]tri]] [/P] [/Q] [/S] [/T[:]heure]] [/W] [/X] [/4]
```

[lecteur:][chemin][nom_de_fichier]

Spécifie le lecteur, le répertoire et/ou fichiers à lister.

/A Affiche les fichiers dotés des attributs spécifiés.

attributs D Répertoires

R Lecture seule

L points d'analyses

H Caché

A Archive

I fichiers indexés sans contenus

S Système

- Préfixe de négation

/B Utilise le format abrégé (noms des fichiers).

/C Affiche le séparateur de milliers pour les tailles de fichiers.

Ceci est la valeur par défaut. Utilisez /-C pour désactiver l'affichage du séparateur.

/D Sur cinq colonnes avec fichiers triés par colonne.

/L Affiche en minuscules.

/N Nouveau format longue liste où les noms de fichiers sont à droite.

/O Affiche les fichiers selon un tri spécifié.

tri N Nom (alphabétique) S Taille (ordre croissant)

E Extension (alphabétique) D Date et heure (chronologique)

G Répertoires en tête - Préfixe en ordre indirect

/P Arrêt après l'affichage d'un écran d'informations.

/R affiche les flux de données alternatifs du fichier

/Q Affiche le nom du propriétaire du fichier.

/S Affiche les fichiers d'un répertoire et de ses sous-répertoires.

/T Contrôle le champ heure affiché ou utilisé dans le tri.

heure C Création

A Dernier accès

W Dernière écriture

/W Affichage sur cinq colonnes.

/X Affiche les noms courts générés pour les noms de fichier non 8.3 car.

Ce format est celui de /N avec le nom court inséré avant le nom long.

S'il n'y a pas de nom court, des espaces seront affichés à la place.

/4 Affiche l'année sur quatre chiffres.

Les commutateurs peuvent être préconfigurés dans la variable d'environnement

DIRCMD. Pour les ignorer, les préfixer avec un trait d'union. Par exemple /-W.

A propos des wildcards/caractères joker :

Si j'ai quelques fichiers au nom incertain en matière de majuscules-minuscules, je puis rédiger en linux un filename « à tatons » tel que

```
filename a "~\wielki\temp\[Ss][Aa][Ss]erie199[0-9]'.dat";
```

```
data bon;infile a;input nom $ cote;run;
```

Inspiré du regexp, en windows, cependant les possibilités sont plus restreintes :

```
filename a "d:\temp\Saserie????.dat"; ou filename a "d:\temp\Saserie*.dat";
```

sont seuls autorisés pour lire des fichiers tels que SASerie1995.dat et Saserie1996.dat

Ceci n'est utilisable qu'en situation d'input bien évidemment.

2.1.3 Le *catname*

L'instruction CATNAME permet de **concaténer explicitement** des catalogues Sas existant sous réserve de préexistence des librefs utilisés. Cette instruction créera une notion de *catref* semblable à celle de *libref* ou *fileref*.

Ce *catref* sera particulièrement utile pour rassembler par exemple virtuellement divers catalogues de formats (ceux généraux à l'Ined, ceux du groupe de travail autour de l'enquête et les vôtres), même si l'option FMTSEARCH= permettait déjà quelque chose de semblable.

```
LIBNAME CGROUP "E:\GIPSANTE\PROG\" ;
LIBNAME PERSO "C:\MONETUD\SASCAT\" ;
LIBNAME TOUS "C:\TEMP\" ; *LIEU FICTIF DU FUTUR CATREF;
CATNAME TOUS.FORMATS (CGROUP.FORMAT PERSO.MONCAT) ;
OPTIONS FMTSEARCH=TOUS.FORMATS ;
```

En situation de lecture/mise à jour, l'utilisation du *catref.nom.entrytype* fera référence à la première entrée de ce nom rencontrée en suivant l'ordre de la concaténation.

En situation d'écriture, il y aura création dans le premier catalogue mentionné de la concaténation, même si une entrée de même nom existe plus loin. Si le premier catalogue n'existe pas en mise à jour, Sas utilisera le suivant. Un delete, rename ou contents ne portera toujours que sur la première occurrence d'une entrée.


Sont autorisés bien sûr:

```
CATNAME LIBREF.CATREF (LIB1.CAT1 (ACCESS=READONLY) LIB2.CAT2) ;
CATNAME LIBREF.CATREF ALL CLEAR ;
CATNAME LIBREF.CATREF LIST ;
```


Mais on peut toujours se contenter d'effectuer une concaténation implicite à l'aide d'un libname.

2.1.4 Gestes en v9

Exploration progressive des Librairies

Un double clic sur l'objet Librairies ouvrira sur la liste des librairies actives. La même opération sur une librairie ouvrira sur la liste des membres de cette librairie. Pour remonter, utiliser  de la ToolBox/Boite à outils.




Pour connaître les caractéristiques d'une librairie (sélectionnée), on activera la rubrique *Properties* par un clic droit.



Le menu déroulant View et ses premières rubriques *Large Icons*, *Small icons*, *List*, *Details*, *Show tree*, *Up one level* et *Refresh* permettent d'adapter l'apparence du contenu de la fenêtre Explorer.

Informations et mouvements sur les tables

Après sélection d'une table, un clic droit avec sélection de Properties, vous conduira vers la vue de différents panneaux informatifs (équivalent du Proc contents).



Après sélection d'une table, un clic droit présente aussi d'autres possibilités :

- Open conduira vers l'ouverture de la table en mode Viewtable ([annexe A1](#))
- View Column reproduira le deuxième écran ci dessus
- Query vous branchera sur la fenêtre du dispositif SQL Query (voir brochure SQL et SAS)
- Export vous donnera accès au dispositif de Sas access to Pc File Format
- View in Excel ouvrira une feuille Excel temporaire dans la work de Sas, grâce au programme ODS suivant :

```

2  filename _temp_ "d:\My SAS Files\work\_TD1432\#LN00009.xls";
3  ods noresults;
4  ods listing close;
5  ods html file=_temp_ rs=none style=minimal;
NOTE: Writing HTML Body file: _TEMP_
6  proc print data=Sashelp.'Class'N label noobs;
7  run;
8  ods html close;
9  ods results;
10 ods listing;
11 filename _temp_;
NOTE: Fileref _TEMP_ has been deassigned.
12 dm "winexecfile " "d:\My SAS Files\work\_TD1432\#LN00009.xls" " "
```

- Save as Html créera un fichier html à l'emplacement par vous souhaité sans l'ouvrir

```

13 filename _temp_ "d:\My SAS Files\9.2\Class.html";
14 ods noresults;
15 ods listing close;
16 ods chtml file=_temp_ rs=none;
NOTE: Writing CHTML Body file: _TEMP_
17 proc print data=Sashep.'Class'N noobs;
18 run;
19 ods chtml close;
20 ods results;
21 ods listing;
22 filename _temp_;
NOTE: Fileref _TEMP_ has been deassigned.

```

- Print effectuera un proc print avec résultat dans l'Output mais n'utilisez pas cette option, car ce n'est absolument pas optimisé et risque d'épuiser rapidement le stock de papier de l'imprimante.
ODS PRINTER;PROC PRINT DATA=Sashep.'Class'N;RUN;ODS PRINTER CLOSE;

- Copy contents to clipboard créera par l'ODS le contenu d'une page html (4.1 Transitional) que vous pouvez sauvegarder à partir du Clipboard.

```

23 filename _temp_clipbrd;
24 ods noresults;
25 ods listing close;
26 ods html file=_temp_ rs=none style=minimal;
NOTE: Writing HTML Body file: _TEMP_
27 proc print data=Sashep.'Class'N noobs;
28 run;
29 ods html close;
30 ods results;
31 ods listing;
32 filename _temp_;
NOTE: Fileref _TEMP_ has been deassigned.

```

- Copy permettra de copier l'icône d'une librairie dans une autre librairie sous réserve des bonnes sélections intermédiaires.

Si par hasard par erreur, vous avez fermé la fenêtre Explorer, sa réouverture se fera différemment sans ce menu détaillé mais avec un menu simplifié moins riche.

2.2 Le fichier de configuration Config

Pour démarrer une session Sas, il est nécessaire de disposer d'un fichier **SASV9.CFG**, fourni par Sas Institute et adapté par votre support Sas local à la situation institutionnelle. Le double clic sur l'icône de Sas Windows en provoque le déclenchement.

En version 9.2.3, nous avons *en fait* 4 types d'ouverture de Sas

- une version **anglaise** qui, bien que réglée sur la langue française et le type de date DDMMYY, présente des menus et autres détails en anglais et par conséquent proche du help de Sas qui reste en Anglais.
-CONFIG "C:\Program Files\SAS\SASFoundation\9.2\nls\en\SASV9.CFG"
- une version **francisée** également accessible par le Menu/Programmes ou par une icône correctement créée
-CONFIG "C:\Program Files\SAS\SASFoundation\9.2\nls\fr\SASV9.CFG"
- une version **utf-8** (cachée en SAS/SAS9.2 ...Utilitaires) permettant de travailler en double byte sur des fichiers unicode
-CONFIG "C:\Program Files\SAS\SASFoundation\9.2\nls\u8\SASV9.CFG"
- l'ouverture **par le clic droit** sur un objet Sas (dans l'Explorateur Windows), qui ouvrira la version francisée du logiciel *-si celui-ci n'est pas déjà ouvert-* de par le passage unique en registre de la clé qui utilise le C:\Program Files\SAS\SASFoundation\9.2\sasv9.cfg contenant un renvoi à la version française !
-CONFIG "C:\Program Files\SAS\SASFoundation\9.2\nls\fr\SASV9.CFG"

Si vous voulez changer ce comportement, mettez-y la référence anglaise dans ce fichier !


Ces modifications sont réservées aux spécialistes et met en jeu leur responsabilité, car Sas risque de ne plus marcher du tout.

Ces fichiers ne seront ouverts **strictement qu'avec un éditeur texte** : le bloc-notes de Windows, le Sas text editor, le Sas universal viewer, Crimson Editor°, Notepad++°, PsPad° ou UltraEdit°, mais n'utilisez jamais Word. (° mentionnés car pouvant présenter ou présentant une coloration syntaxique pour Sas et d'autres avantages)

A l'exception du fichier à la racine du répertoire de la version 9.2, les autres se composent de deux parties : les options SYSTEME et la partie - INSTALL (après l'avertissement /*Do not edit below this line - INSTALL application */).

L'utilisateur courant se sert du *sasv9.cfg* fourni ; éventuellement, il copie celui-ci dans un répertoire sien, sous réserve de modifier les propriétés de **son** icône correctement.

En ce cas, l'appel de ce fichier se présentera de la façon suivante :



← ne jamais laisser %USERPROFILE% ici car écriture alors sur c:\ ...

Et c'est seulement alors qu'il pourra modifier le contenu de son fichier sasv9.cfg

Mais il peut aussi y avoir une recherche automatique du CONFIG

- d'abord dans le répertoire courant (démarrer dans...)
- dans le répertoire défini en !sasroot (=une variable d'environnement SAS)
- ensuite, si des chemins ont été définis en Dos (par ex PATH C:\SASDIRECT; D:\SAS), successivement dans les répertoires mentionnés
- et enfin, à la racine du disque courant


```

-SET sasext0 "C:\Program Files\SAS\SASFoundation\9.2"
-SET sasroot "C:\Program Files\SAS\SASFoundation\9.2"
-SET sasext1 "C:\Program Files\SAS\SASFoundation\9.2\nls"

/* Setup the MYSASFILES system variable */
-SET MYSASFILES "d:\My SAS Files\9.2\"

/* Setup the default SAS System user profile folder */
-SASUSER "d:\My SAS Files\9.2\sasuser"

/* Setup the default SAS System user work folder */
-WORK "d:\My SAS Files\work"

/* Setup the SAS System configuration folder */
-SET SASCFG "C:\Program Files\SAS\SASFoundation\9.2\nls\en"

/* location of help in OS help format */
-HELPLoc ("!MYSASFILES\classdoc" "!sasroot\nls\en\help" "!sasroot\core\help")

/* Location for Java applets */
-APPLETLOC "C:\Program Files\SAS\SASGraphJavaApplets\9.2"

/* Location for SAS Textures */
-TEXTURELOC !sasroot\common\textures

/* Options used when SAS is accessing a JVM for JNI processing */
-JREOPTIONS=(-Dsas.jre.libjvm=C:\PROGRA~1\Java\JRE15-1.0_1\bin\client\jvm.dll -
Djava.security.policy=!SASROOT\core\sasmisc\sas.policy -
Dsas.ext.config=!SASROOT\core\sasmisc\sas.java.ext.config -
Dsas.app.class.path=C:\PROGRA~1\SAS\SASVER~1\9.2\eclipse\plugins\tkjava.jar -
DPFS_TEMPLATE=!SASROOT\core\sasmisc\qrpfstpt.xml -
Djava.class.path=C:\PROGRA~1\SAS\SASVER~1\9.2\eclipse\plugins\SASLAU-1.JAR -
Djava.system.class.loader=com.sas.app.AppClassLoader -Xmx128m -Xms128m -
Djava.security.auth.login.config=!SASROOT\core\sasmisc\sas.login.config -
Dtkj.app.launch.config=!SASROOT\picklist)

/* Set Windows Locale */
-LOCALE "fr_FR"

/* Default resources location */
-RESOURCESLOC ("!sasroot\nls\en\resource" "!sasroot\core\resource")

/* SAS/CONNECT Software script files */
-SASSCRIPT (!sasext0\connect\saslink)

/* graph/maps */
-MAPS !sasext0\maps
Etc etc etc

```

Commentaires explicatifs sur cette configuration :

Les lignes **en gras** sont liées à divers réglages de Sas à l'INED visant à éviter l'écriture sur C:\ et uniformiser son implantation en vue d'un meilleur dépannage.

Le paramètre **-SASINITIALFOLDER "d:\My SAS Files\9.2"** *s'il est activé et modifié* sert à spécifier le lieu par défaut des *open* et *save* des boîtes de dialogue (autre que le répertoire *sasuser*) : vous pouvez le modifier en ce sens en vue d'une personnalisation.

Les paramètres **-EMAILSYS SMTP -EMAILHOST webmail.ined.fr -EMAILPORT 25** autorisent la programmation de emails par code.

Les **HELP REGISTER** permettent de faire apparaître dans le menu Help de l'espace de travail Sas, une possibilité d'ouvrir des chemins de documentation en ligne.

```

/* Set Windows Locale */-LOCALE "fr_FR" en cfg english mais -LOCALE "French" en cfg français
et -DBCS -ENCODING UTF-8 pour le cfg u8!

```


2.4 Le catalogue Profile de la Sasuser

Comme sur le serveur avec Sas LINUX, le catalogue Profile de la librairie SASUSER sous Windows assure la mémoire des habitudes de l'utilisateur du poste en matière de fenêtrage, palette de couleurs, bref, de style de travail. Sas le cherchera dans la librairie \SASUSER.

Si vous êtes à deux à travailler sur un même poste et que vos habitudes sont conflictuelles, appelez votre propre *profile* en joignant au SAS.EXE -SASUSER D:\SASANDRE. Ceci suppose l'existence du répertoire SASANDRE et le fait qu'il contienne au moins les fichiers PROFILE.SAS7BCAT et PROFILE2.SAS7BCAT que vous pouvez y copier.

L'option d'appel -Rsasuser "... " est aussi utile. -Verbose parfois aussi. *Venir me voir s'il vous plait.*

Pour faciliter les mises à jour du logiciel, il est indispensable de ne pas écrire des fichiers personnels dans ces catalogues système, ni même dans le répertoire référencé dans le -sasuser du config. En effet, en cas de crash et de réinstallation, le répertoire d:\My SAS Files\9.2\sasuser n'est jamais sauvegardé, ni réutilisé. Il est recréé à neuf.

D'autres fichiers sont également présents dans la Sasuser et servent aux réglages du logiciel : e. a. REGISTRY.SAS7BITM, PARM.SAS7BDAT, TOOLS.SAS7BCAT et TEMPLAT.SAS7BITM.

2.5 La Work ou librairie de travail temporaire


Comme sur toute version Sas, un espace disque temporaire est sollicité ici aussi en Windows. La librairie WORK est affectée par défaut à un répertoire de travail attribué par le fichier de configuration. Les fichiers temporaires qui y seront écrits, seront effacés après le BYE ou l'Exit de SAS, car le logiciel nettoie le sous-répertoire \WORK, **sauf en cas de plantage**. A partir de la version 9.2.3, la work de SAS à l'INED est définitivement en D:\My SAS File\work quelque soit la version future de Sas.

C'est pour cela qu'il est **important de bien fermer le logiciel !**

Si malgré tout il faut la nettoyer, faites-le ! Il s'agit de supprimer les répertoires de type TD##### ; mais en général, vous tombez sur un répertoire work actif et vous hésitez. Opérez donc SAS fermé OU...

Sas, en 9.2, offre la possibilité en Windows de le faire autrement par le Disk Cleaner Handler.

Lancez la commande système *cleanmgr*




The image shows two overlapping Windows windows. The left window is titled 'Exécuter' (Run) and contains the text 'Entrez le nom d'un programme, dossier, document ou d'une ressource Internet, et Windows l'ouvrira pour vous.' Below this is a text box with 'cleanmgr' entered and a dropdown arrow. The right window is titled 'Nettoyage de disque pour Données (D:)' (Disk Cleanup for Data (D:)). It shows a progress bar and text: 'Vous pouvez utiliser le Nettoyage de disque pour libérer jusqu'à 1 344 685 Ko d'espace disque sur Données (D:).' Below this is a table of files to be deleted:

Fichier à supprimer :	Taille
<input type="checkbox"/> Fichiers d'installation d'Office	0 Ko
<input checked="" type="checkbox"/> Corbeille	149 586 K
<input checked="" type="checkbox"/> SAS Temporary Files	10 662 Ko
<input type="checkbox"/> Compression des fichiers non utilisés	1 184 439
<input type="checkbox"/> Fichiers catalogue d'indexation du contenu	0 Ko

Below the table, it says 'Espace total gagné : 149 586 K'. There is a 'Description' section with the text 'Removes temporary files and directories created by SAS.' At the bottom are 'View Files...', 'OK', and 'Annuler' buttons.

Répondez D:\ pour la localisation des fichiers Sas temporaires à effacer
Après un certain temps nécessaire à l'examen du disque, il présente les possibilités.
Ayant choisi Sas Temporary Files, faites View Files



2.6 L'accès au système d'exploitation et le multitâche

Pour accéder au système à l'intérieur d'une session Sas interactive, la commande X est toujours nécessaire. Son fonctionnement est conditionné par la valeur de deux options Sas (xwait et xsync)

- avec XWAIT et XSYNC (options par défaut), la commande X saisie seule permet d'accéder au Dos (on se retrouve dans le répertoire ouvert courant) et nécessite en fin de parcours la saisie de la commande EXIT pour retourner à Sas. X suivi d'une commande système sollicitera après l'exécution de la commande, suivie par un message ("press any key to return to Sas"), une action concrète supplémentaire.
- avec NOXWAIT, la commande X *argument Dos* sera automatiquement fermée après exécution et nous nous retrouverons dans Sas.
- avec NOXSYNC, l'autre tâche ouverte est indépendante de la session Sas, qui reste active et utilisable.

On mettra aussi souvent une troisième option XMIN pour maintenir le *sessionDos* dans un état de fenêtrage minimal. Un exemple a déjà été présenté page 16 Exemple 7.

Si vous connaissez le macro-langage, il y a aussi une possibilité intéressante avec le %sysexec :

```
ods listing close;
ods rtf file="d:\temp\test.rtf";
ods tagsets.excelxp file="essai.xls.xml";
proc report nowd data=sashelp.class; where name='I'; run;
ods tagsets.excelxp close;
%sysexec essai.xls.xml;
ods rtf close;
```

Le multitâche possible de Windows est supporté par Sas. Mais on s'interdira l'ouverture de plusieurs sessions Sas en même temps.